



CUSTOMER EDUCATION SERVICES

SystemVerilog Testbench Workshop

Lab Guide

50-I-052-SLG-016 2019.06

Synopsys Customer Education Services
690 E. Middlefield Road
Mountain View, California 94043

Workshop Registration: <https://training.synopsys.com>

Copyright Notice and Proprietary Information

© 2019 Synopsys, Inc. All rights reserved. This software and documentation contain confidential and proprietary information that is the property of Synopsys, Inc. The software and documentation are furnished under a license agreement and may be used or copied only in accordance with the terms of the license agreement. No part of the software and documentation may be reproduced, transmitted, or translated, in any form or by any means, electronic, mechanical, manual, optical, or otherwise, without prior written permission of Synopsys, Inc., or as expressly provided by the license agreement.

Destination Control Statement

All technical data contained in this publication is subject to the export control laws of the United States of America. Disclosure to nationals of other countries contrary to United States law is prohibited. It is the reader's responsibility to determine the applicable regulations and to comply with them.

Disclaimer

SYNOPSYS, INC., AND ITS LICENSORS MAKE NO WARRANTY OF ANY KIND, EXPRESS OR IMPLIED, WITH REGARD TO THIS MATERIAL, INCLUDING, BUT NOT LIMITED TO, THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE.

Trademarks

Synopsys and certain Synopsys product names are trademarks of Synopsys, as set forth at <http://www.synopsys.com/Company/Pages/Trademarks.aspx>. All other product or company names may be trademarks of their respective owners.

Third-Party Links

Any links to third-party websites included in this document are for your convenience only. Synopsys does not endorse and is not responsible for such websites and their practices, including privacy practices, availability, and content.

Synopsys, Inc.
690 E. Middlefield Road
Mountain View, CA 94043
www.synopsys.com

Document Order Number: 50-I-052-SLG-016
SystemVerilog Testbench Workshop Lab Guide

1

SystemVerilog Verification Flow

Learning Objectives

After completing this lab, you should be able to:

- Create the **SystemVerilog** testbench files for a Device Under Test (DUT)
- Write a SystemVerilog **task** to reset the DUT
- **Compile and simulate** the SystemVerilog test program
- Verify that the DUT signals are driven as specified with Discovery Verification Environment (dve) or Verdi Automated Debug Platform



Lab Duration:
30 minutes

Lab 1

Getting Started

Once logged in, you will see three directories: `rtl`, `labs` and `solutions`.




Figure 1. Lab Directory Structure

In this lab, you will develop a simple SystemVerilog test program to reset the DUT.




Figure 2. Testbench Architecture

Lab Overview

This lab takes you through the process of building, compiling, simulating and debugging the testbench:





Figure 3. Lab 1 Flow Diagram

Note: You will find answers for all questions and solutions in the Answers / Solutions section at the end of this lab.


Constructing SystemVerilog Testbench

A typical architecture of a SystemVerilog testbench looks like the figure below:



The process with VCS in creating this SystemVerilog testbench is as follows:

- Create an interface to connect the test program and the DUT.
- Write test program(s).
- Connect the test and DUT using a harness file including the interface.



Task 1. Logging In

1. Log into your workstation if needed (use the login and password provided by the instructor).
2. Go to the **lab1** directory:

```
> cd labs/lab1
```

Task 2. Create SystemVerilog Interface File

1. There is a skeleton **router_io.sv** file. Open it with a text editor.
- Note:** The skeleton files in the labs have comments as markers, indicated by a “ToDo” to guide you through the lab steps.
2. The signals needed to connect to the DUT are already entered for you:

```
interface router_io(input bit clock);
    logic reset_n ;
    logic [15:0] din ;
    ...
    logic [15:0] frameo_n ;
endinterface: router_io
```

At this stage, all interface signals are asynchronous and without a direction specification (i.e. input, output, inout). The direction can only be specified in a clocking block for synchronous signals or a modport for asynchronous signals.

The next step is to create the set of synchronous signals for the test program to drive and sample the DUT signals. This is done with **clocking** block declarations.

Lab 1

3. Declare a clocking block driven by the posedge of the signal **clock**. This clocking block will be used by the test program to execute synchronous drives and samples. All directions for the signals in this clocking block must be with respect to the test program.

```
interface router_io(input bit clock);  
    logic reset_n;  
    logic [15:0] din;  
    ...  
    logic [15:0] frameo_n;  
    clocking cb @(posedge clock);  
        output reset_n;  
        output din; // no bit reference  
        output frame_n; // no bit reference  
        output valid_n; // no bit reference  
        input dout; // no bit reference  
        input valido_n; // no bit reference  
        input busy_n; // no bit reference  
        input frameo_n; // no bit reference  
    endclocking: cb  
endinterface: router_io
```

Create synchronous
signals by placing
signals into clocking
block with directions
specific to test
program

4. If desired, add specifications for input and output skews (output recommended):

```
interface router_io(input bit clock);  
    ...  
    clocking cb @(posedge clock);  
        default input #1ns output #1ns;  
        output reset_n;  
        output din; // no bit reference  
        ...  
    endclocking: cb  
endinterface: router_io
```

5. Lastly, create a modport to be used for connection with the test program. In the argument list, there should be references to the clocking block created in the previous step and all other potential asynchronous signals.

```

interface router_io(input bit clock);
    ...
    clocking cb @(posedge clock);
        default input #1 output #1;
        output reset_n;
        output din;      // no bit reference
    ...
endclocking: cb
modport TB(clocking cb, output reset_n);
endinterface: router_io

```

6. Save and close the file.

Task 3. Create SystemVerilog Test Program File

1. Open the SystemVerilog test program file called **test.sv** with an editor
2. In this file, declare a test program block with arguments which connects to the TB modport in the interface block:

```

program automatic test(router_io.TB rtr_io);

endprogram: test

```

3. Within the program block, print a simple message to the screen:

```

program automatic test(router_io.TB rtr_io);
    initial begin
        $display("Hello World!");
    end
endprogram: test

```

4. Save and close the file.

Lab 1

Task 4. Create SystemVerilog Test Harness File

1. Open the skeleton **router_test_top.sv** provided with a text editor.
2. It looks like

```
module router_test_top;
    parameter simulation_cycle = 100;
    bit SystemClock = 0;
    router dut(
        .reset_n  (reset_n),
        .clock    (clock),
        .din      (din),
        .frame_n  (frame_n),
        .valid_n  (valid_n),
        .dout     (dout),
        .valido_n (valido_n),
        .busy_n   (busy_n),
        .frameo_n (frameo_n)
    );
    ...
    always begin
        #(simulation_cycle/2) SystemClock =
            ~SystemClock;
    end
endmodule
```

3. Add an interface instance to the harness:

```
module router_test_top;
    parameter simulation_cycle = 100;
    bit SystemClock = 0;
    router_io top_io(SystemClock); ←
    router dut( ... );
    ...
    always begin
        #(simulation_cycle/2) SystemClock =
            ~SystemClock;
    end
endmodule
```

Instantiate

4. Instantiate the test program (make I/O connection via interface instance):

```
module router_test_top;
  parameter simulation_cycle = 100;
  bit SystemClock = 0;
  router_io top_io(SystemClock); // interface
instance
  test t(top_io); // program instance
  router dut( ... );
...
always begin
  #(simulation_cycle/2) SystemClock =
~SystemClock;
end
endmodule
```

5. Modify DUT connection to connect via interface:

```
router dut(
  .reset_n  (top_io.reset_n),
  .clock    (top_io.clock),
  .din      (top_io.din),
  .frame_n  (top_io.frame_n),
  .valid_n  (top_io.valid_n),
  .dout     (top_io.dout),
  .valido_n (top_io.valido_n),
  .busy_n   (top_io.busy_n),
  .frameo_n (top_io.frameo_n)
);
```

Connect DUT via
interface instance

6. Add

- ``timescale`
- `$timeformat` (in `initial` block)

```
`timescale 1ns/100ps
module router_test_top;
...
initial begin
$timeformat(-9, 1, "ns", 10);
end
...
endmodule
```

Set time (%t)
to ns scale

7. Save and close the file


Lab 1

Task 5. Compile & Simulate

1. **Compile** all files with the following command (using vcs):

```
> vcs -sverilog router_test_top.sv test.sv router_io.sv ../../rtl/router.v
```

VCS will compile the files and create an executable file called **simv**.



The next step is to execute the simulation binary.

2. Run the simulation by executing the binary created by VCS:


```
> simv
```

3. Check to see if you see the **\$display()** message. If so, proceed to the next step. If not, debug your testbench to see why the message was not displayed.

Task 6. Reset the Router

You have successfully built, compiled and simulated a simple SystemVerilog testbench. The next step is to add needed functionality to the testbench. In this task, you will add a routine to reset the router.

1. Open **test.sv** with a text editor.
2. In the **program** block define a task called **reset()** to reset the DUT per spec. as described in lecture: (bear in mind that **reset_n** can be either a synchronous or an asynchronous signal)



3. In the initial block, replace **\$display()** with a call to **reset()**

When completed, your program may look something like:

```
program automatic test(router_io.TB rtr_io);
    initial begin
        reset();
    end
    task reset();
        rtr_io.reset_n = 1'b0;           // asynchronous
        rtr_io.cb.frame_n <= '1;       // synchronous
        rtr_io.cb.valid_n <= '1;       // synchronous
        repeat(2) @(rtr_io.cb);        // synchronous
        rtr_io.cb.reset_n <= 1'b1;     // synchronous
        repeat(15) @(rtr_io.cb);       // synchronous
    endtask: reset
endprogram: test
```

4. Save and close the file.

Task 7. Compile & Simulate

Once the test code is completed, compile and simulate the DUT code with the SystemVerilog testbench.

1. If you wish to use the Verdi debugging environment, add the instrumentation for Verdi in router_test_top.sv **initial** block. You can use Verdi as detailed later in Task 9. This requires the **-kdb** and **-lca** switches. It creates the dump file “**novas.fsdb**” which is used by Verdi waveform viewer.

```
initial begin
    $fsdbDumpvars;      // Only for Verdi debugging
```

2. **Compile** all files with the following command (using vcs):

```
> vcs -sverilog -debug_access+all +vcs+vcdpluson router_test_top.sv \
test.sv router_io.sv ../../rtl/router.v -kdb -lca
```

The above compile added usage of some more switches:

-debug_access+all, +vcs+vcdpluson, -kdb, -lca

These first two switches will create a dump file “**vcdplus.vpd**” which stores the activities of signals during simulation. This file is used by VCS Discovery Visual Environment (DVE) to examine the DUT signals in a waveform window.

Lab 1

3. Run the simulation by executing the binary created by VCS.

```
> simv
```

Did the simulation execute correctly? Looking at the simulation print out, there is no way to know for sure. You will need to examine the simulation waveform with a waveform viewer to verify that your testbench was executed correctly.

For future compile/simulate iterations, there is a make file (**Makefile**) already written to simplify this process. To use it, just type “**make**” to recompile and run simulation.

The content of the **Makefile** is as follows :

(Type **make help** to get a summary description of content)

```
# Makefile for SystemVerilog Testbench Lab1
LAB_DIR = lab1
RTL= ../../rtl/router.v
SVTB = ./router_test_top.sv ./router_io.sv ./test.sv
seed = 1
run_opts =
comp_opts =

default: test

test: compile run

run:
    ./simv -l simv.log +ntb_random_seed=$(seed) $(run_opts)

compile:
    vcs -l vcs.log -sverilog -kdb -debug access+all $(SVTB) $(RTL)
+vcs+vcpluson -lca -full64 $(comp_opts)

verdi:
    verdi -ssf novas.fsdb -undockWin -nologo &

dve:
    dve -vpd vcdplus.vpd &

verdi_debug:
    ./simv -l simv.log -gui=verdi +ntb_random_seed=$(seed) $(run_opts)

debug:
    ./simv -l simv.log -gui=dve +ntb_random_seed=$(seed) $(run_opts)

solution: clean
    cp -f ../../solutions/${LAB_DIR}/*.sv .

clean:
    rm -rf simv* csrv* *.tmp *.vpd *.key *.log *hdrs.h *.fsdb verdiLog
elabcomLog novas.* *.dat

original: nuke
    cp -f ../../solutions/${LAB_DIR}/router_io.sv.orig router_io.sv
    cp -f ../../solutions/${LAB_DIR}/router_test_top.sv.orig
router_test_top.sv
    cp -f ../../solutions/${LAB_DIR}/test.sv.orig test.sv

nuke: clean
    rm -rf *.v* *.sv include .*.*.lock .*.*.old DVE* *.tcl *.h *.rc *.ses
*.ses.* *.dat

help:
```

Task 8. Browse Simulation Hierarchy in DVE


Note: If you wish to use Verdi instead of DVE for viewing waveforms, please skip to Task 9. **Do so only if you are familiar with Verdi Automated Debug Platform.**

1. Bring up the waveform viewer using the UNIX command **dve**.


```
> dve &
```

The GUI may not look exactly like the images shown below.

2. Click on **File → Open Database**.



3. Double click on **vcdplus.vpd** file.




The DVE debugging windows should now be opened with the harness hierarchy and source code.


To see the waveforms, you will need to open a Waveform window.

Lab 1


4. Expand harness to access interface signals by clicking on .




5. Click interface instance to highlight, then click on to display waveform.




6. The Waveform window should open looking like the following:



7. Set waveform time scale to 1ns



8. Click on to span signal window to see through all simulation time.




9. Verify the correct timing of the reset signal. If there are errors, correct the error then type “**make**” to re-compile and re-simulate the modified code.
10. You can save the waveform setting. In the Wave window select File -> Save Current View menu command. In the pop-up dialog box, save the session as **router_waves.tcl**. Make sure to save this file again whenever you change the waveform settings.
11. Reuse the setting with File → Recent Session →/lab1/router_waves.tcl in future labs after loading the database as shown in step 2 above.

Lab 1

Task 9. Browse Simulation Hierarchy in Verdi


Note: Do this step only if you wish to use Verdi Automated Debug Platform waveform viewer to debug the workshop labs.

1. Open the Verdi waveform viewer. It may take a while for the GUI to appear. The **novas.fsdb** file is loaded in the waveform tool by the command
> make verdi
2. You should see two windows – the main Verdi console nTrace and the waveform window nWave - as shown below.




Continued...

3. The testbench hierarchy is shown on the top left of the nTrace window. You can use the to expand and the to collapse the hierarchy.



4. Using the Middle Mouse Button (MMB), drag the top_io from the Instance pane to the nWave Window. This opens the interface signals in the nWave window as shown below.
5. You can resize the different sections of the window as desired.




6. Change the time unit to 1ns. Click on the 100 ps. A separate box pops up to let you set the time unit. Set it to 1ns as shown.



To display complete simulation time, click on the icon.

Lab 1



7. Verify the correct timing of the reset signal. If there are errors, correct the error then type the correct “**make**” commands to re-compile and re-simulate the modified code. Zoom in on the area where signal **reset_n** goes inactive and check the 1 ns drive delay you coded in the clocking block.

If there are no errors, you are done with lab 1.

Answers / Solutions

router io.sv Solution:

```
interface router_io(input bit clock);
    logic             reset_n;
    logic [15:0]      din;
    logic [15:0]      frame_n;
    logic [15:0]      valid_n;
    logic [15:0]      dout;
    logic [15:0]      valido_n;
    logic [15:0]      busy_n;
    logic [15:0]      frameo_n;

    clocking cb @(posedge clock);
        default input #1ns output #1ns;
        output reset_n;
        output din;
        output frame_n;
        output valid_n;
        input  dout;
        input  valido_n;
        input  busy_n;
        input  frameo_n;
    endclocking: cb

    modport TB(clocking cb, output reset_n);
endinterface: router_io
```

test.sv Solution:

```
program automatic test(router_io.TB rtr_io);

    initial begin
        reset();
    end

    task reset();
        rtr_io.reset_n <= 1'b0;
        rtr_io.cb.frame_n <= '1;
        rtr_io.cb.valid_n <= '1;
        repeat(2) @(rtr_io.cb);
        rtr_io.cb.reset_n <= 1'b1;
        repeat(15) @(rtr_io.cb);
    endtask: reset

endprogram: test
```

router test top.sv Solution:

```
`timescale 1ns/100ps
module router_test_top;
    parameter simulation_cycle = 100;

    bit SystemClock = 0;
    router_io top_io(SystemClock);
    test t(top_io);
    router dut(
        .reset_n      (top_io.reset_n),
        .clock (top_io.clock),
        .din   (top_io.din),
        .frame_n     (top_io.frame_n),
        .valid_n     (top_io.valid_n),
        .dout  (top_io.dout),
        .valido_n    (top_io.valido_n),
        .busy_n      (top_io.busy_n),
        .frameo_n    (top_io.frameo_n)
    );

    initial begin
        $timeformat(-9, 1, "ns", 10);
        $fsdbdumpvars //only for Verdi debugging
    end

    always begin
        #(simulation_cycle/2) SystemClock = ~SystemClock;
    end

endmodule
```

2

Sending Packets Through Router

Learning Objectives

After completing this lab, you should be able to:

- Extend the SystemVerilog testbench from lab1 to send a packet from an input port through an output port
- Compile and simulate the router with the updated SystemVerilog testbench
- Verify stimulus generation and driver signals with DVE or Verdi3



Lab Duration:
60 minutes

Lab 2

Getting Started

In lab1, you have familiarized with the process of building the simulation environment for verifying a DUT with SystemVerilog.

In this lab, you will continue to build the next component of the testbench: stimulus generator, protocol transactors and device drivers. You will develop a set of routines to drive one packet into input port 3 and visually observe the payload of this packet coming out of output port 7.




Figure 1. Lab 2 testbench architecture

Lab 2-2

Sending Packets Through Router
Synopsys SVTB Workshop

Lab Overview

The process of updating, compiling, simulating and debugging the testbench:




Figure 2. Lab 2 Flow Diagram

Note: You will find Answers for all questions and solutions in the Answers / Solutions at the end of this lab.

Lab 2

Sending a Packet through the Router

Task 1. Copy Files from the Previous Directory

To keep everyone in the workshop synchronized in developing the SystemVerilog testbench code, you will adopt a strategy of basing all lab work on standard solutions from previous labs. For lab 2, you will copy the files from the lab1 solutions directory, and develop your lab 2 testbench based on these files.

1. CD into the lab2 directory.

```
> cd ..\lab2
```

2. Copy the source files in the **solutions/lab1** directory into the current directory with **make** script.

```
> make copy
```

Note: If you chose to use your own lab files from lab1, type “**make mycopy**”. However, your files will not have the ToDo markers to guide you.

Task 2. Declaring Program Global Variables

Sending a **packet** through the router requires specifying which **input port** and **output port** to use, and what **data** to send.


To make referencing these variables simple, you will declare them as program global variables.

1. Open **test.sv** with text editor.
2. Declare the **program global variables** for the packet:

```
bit [3:0] sa;           // source address (input port)
bit [3:0] da;           // destination address (output port)
logic[7:0] payload[$]; // packet data array
                      // logic stores 0, 1, x and z values
```

Task 3. Generate Packet Data

In Lab 1, you configured the DUT by calling the `reset()` subroutine. You will continue the testbench development by creating a subroutine to generate the test stimulus in a task called “`gen()`”.



1. In the `program initial` block, call the generator, `gen()`, after `reset()`
2. Following the task `reset()` code block, declare task `gen()`:

```

program automatic test(router_io.TB rtr_io);
    bit [3:0] sa;           // source address
    bit [3:0] da;           // destination address
    logic [7:0] payload[$]; // packet data array

    initial begin
        reset();
        gen();
    end

    task reset();
        ...
    endtask: reset

    task gen();
    endtask: gen
endprogram: test

```

3. In the body of the `gen()` task:
 - a) Set `sa` to 3 and `da` to 7.
 - b) Fill the `payload` queue with 2 to 4 random bytes.

Lab 2

When completed, your code should look something like:

```
task gen();
    sa = 3;
    da = 7;
    payload.delete();
    repeat($urandom_range(4,2))
        payload.push_back($urandom);
    endtask: gen
```

Task 4. Create Send Packet Routines

With the packet information generated, you are now ready to develop the transactor and device-driver routines to send a packet through the router.


Sending a packet through the router is done using three processes(waveform on next page):

Send the:


- Destination Address
- Padding bits
- Payload

Each process will be developed as an individual device driver routine. A transactor routine will call these device driver routines to execute a complete transaction.

The distinction between device drivers and transactor is that device drivers interact with the hardware signals directly while the transactor calls the device drivers. These layers of abstraction make the testbench routines more manageable, portable and reliable.



The following steps take you through the development of these routines.



1. In the initial block, call **send()** task to send a packet immediately after the **gen()** statement.
2. Then, advance simulation time by 10 additional clock cycles after the **send()** call.

This will allow the data coming out of the router to be observed. Otherwise, when you view the waveform in DVE, the output will appear to be cut off. Verdi3 automatically extends the last set of signal waveforms to enhance viewability.

3. Add a declaration of **send()** task in the program block.
4. In the body of the **send()** task, code the following operations:

- Call **send_addr()**
- Call **send_pad()**
- Call **send_payload()**

5. Create the **send_addr()** task.

This is the device driver that drive the four bits of address into the router.

6. In the body of the **send_addr()** task, code the following operations:
 - Drive the **frame_n** signal as per router specification
 - Drive the **din** signal with the destination address (LSB first)

Recall that **din** is a single-bit serial signal. Use a loop construct to drive **din** one bit per clock cycle for four cycles. Each port is represented by an individual bit in **frame_n**, **din**, and **valid_n**. Make sure you specify the correct bit.

Lab 2

Driving the **frame_n** signal for input port 3 to “1” takes the form of:
rtr_io.cb.frame_n[3] <= 1'b1;

Question 1. What will this driving statement do?

rtr_io.cb.frame_n <= 1;

.....
.....

7. Create the **send_pad()** task.

This is the device driver that drives the five padding bits into the router.

8. In the body of the **send_pad()** task, code the following operations:

- Drive the **frame_n**, **valid_n** and **din** signals as per router specification

9. Create the **send_payload()** task.

This is the device driver that sends the payload (data) into the router.

10. In the body of the **send_payload()** task, code the following operations:

- Write a loop to execute for each element of **payload**.
- Within this loop, each 8-bit datum of the **payload[\$]** array should be transmitted one bit per clock cycle starting with the lsb.
- Also remember to drive the valid bit, **valid_n**, as per the router specification.
- For the last valid bit of the packet, make sure to set the **frame_n** signal to 1'b1 as per router specification.
- Return the **valid_n** signal back to 1'b1 after the packet completes.

*** Warning ***

Pay particular attention to how you advance the clock. If in one subroutine you advance the clock upon exiting the subroutine, then in another subroutine you advance the clock upon entering the subroutine, erroneous timing may result.

11. Save and close the file.

Task 5. Compile And Debug The Program

1. Compile and simulate the SystemVerilog testbench with **make**.
2. If you see any runtime errors, you must correct them now.
3. If there are no runtime errors, check the operation using DVE or Verdi3.

When you use DVE from this point onwards, you will want to look at individual port signals. The easiest way to do this is to drag the individual bit you want to see into a new group in the Waveform window.

To open DVE Automated Debug use

```
> make dve
```

To open Verdi Automated Debug use

```
> make verdi
```

Task 6. Extend The Program To Send 21 Packets

1. Modify the program block to send 21 packets through the router using the same values for sa and da of 3 and 7 respectively.
2. Compile, simulate & verify using DVE or Verdi3.

Congratulations, you have completed lab2!

Answers / Solutions

Task 1. Create Send Packet Routines

Question 1. What will this driving statement do?
`rtr_io.cb.frame_n <= 1;`

This will drive all 16 input port's frame_n signal. Port 0's frame_n signal will be driven to "1". Ports 1 – 15's frame_n signal will be driven to "0".

If the intention is to drive all 16 input ports to "1" the correct statement would be:

`rtr_io.cb.frame_n <= '1;`

If the intention is to drive an individual port, then the input port number must be specified as a bit selection in the rtr_io.frame_n signal.

e.g. driving input port 5's frame_n signal to "0" takes the following form:

`rtr_io.cb.frame_n[5] <= 1'b0;`

test.sv Solution:

```
program automatic test(router_io.TB rtr_io);
    int run_for_n_packets;      // number of packets to test
    bit[3:0] sa;                // source address
    bit[3:0] da;                // destination address
    logic[7:0] payload[$];     // packet data array

    initial begin
        $vcddpluson;
        run_for_n_packets = 21;
        reset();
        repeat(run_for_n_packets) begin
            gen();
            send();
        end
        repeat(10) @rtr_io.cb;
    end

    task reset();
        rtr_io.reset_n <= 1'b0;
        rtr_io.cb.frame_n <= '1;
        rtr_io.cb.valid_n <= '1;
        repeat(2) @rtr_io.cb;
        rtr_io.cb.reset_n <= 1'b1;
        repeat(15) @rtr_io.cb;
    endtask: reset

    task gen();
        sa = 3;
        da = 7;
        payload.delete();
        repeat($urandom_range(4,2))
            payload.push_back($urandom);
    endtask: gen

    task send();
        send_addrs();
        send_pad();
        send_payload();
    endtask: send
```

Continued...

```
...Continued from previous page
task send_addrs();
    rtr_io.cb.frame_n[sa] <= 1'b0;
    for(int i=0; i<4; i++) begin
        rtr_io.cb.din[sa] <= da[i];
        @rtr_io.cb;
    end
endtask: send_addrs
task send_pad();
    rtr_io.cb.frame_n[sa] <= 1'b0;
    rtr_io.cb.valid_n[sa] <= 1'b1;
    rtr_io.cb.din[sa] <= 1'b1;
    repeat(5) @rtr_io.cb;
endtask: send_pad

task send_payload();
    foreach(payload[index]) begin
        for(int i=0; i<8; i++) begin
            rtr_io.cb.din[sa] <= payload[index][i];
            rtr_io.cb.valid_n[sa] <= 1'b0;
            rtr_io.cb.frame_n[sa] <=
                (index == (payload.size()-1)) && (i==7);
            @rtr_io.cb;
        end
    end
    rtr_io.cb.valid_n[sa] <= 1'b1;
endtask: send_payload

endprogram: test
```

3

Self-Checking

Learning Objectives

After completing this lab, you should be able to:

- Develop a monitor to sample the output of the router
- Develop a checker to verify the output of the router
- Run driver and monitor routines concurrently
- Verify the self-checking mechanism by executing the testbench against a faulty DUT



Lab Duration:
90 minutes

Getting Started

In Lab 3, you will add in the monitors and self-check mechanisms. The architecture is shown below:




Figure 1. Lab 3 testbench architecture

Lab Overview




Figure 2. Lab 3 Flow Diagram

Note:

You will find Answers for all questions and solutions in the Answers / Solutions at the end of this lab.

Self-Checking & Concurrency

Task 1. Copy Files From Lab 3's Solutions Directory

For consistency, you will copy the files from `../.../solutions/lab3` into the `lab3` directory.

1. Go into the lab3 directory.

```
> cd ../../solutions/lab3
```

2. Copy the source files in the `solutions/lab3` directory into the current directory with `make` script.

```
> make copy
```

(If you chose to use your own lab files from lab2, type “`make mycopy`”.)

Task 2. Build The Top-Level Test Environment

In the following steps, you will complete building of the top-level test environment.


1. Edit `test.sv` file.
2. Add a global declaration for an 8-bit (`logic [7:0]`) queue named:
`pkt2cmp_payload[$]`
This queue will be used to store the data sampled from the DUT.
3. For self-checking, modify the program to execute a receive routine `recv()` concurrently with `send()` followed by a self-checking routine `check()`.

```
program automatic router_test(router_io.TB rtr_io);
  ...
  logic[7:0] pkt2cmp_payload[$];

  initial begin
    ...
    repeat(run_for_n_packets) begin
      gen();
      fork
        send();
        recv();
      join
      check();
    end
    repeat(10) @rtr_io.cb;
  end
  ...
endprogram
```

Task 3. Develop Monitor Routines

In this step, you will build the monitor transactor called `recv()`.



This `recv()` transactor will call device monitor `get_payload()` to retrieve a packet payload from the router. This payload shall be stored in `pkt2cmp_payload[$]` queue by the `get_payload()` routine.

1. Declare the `recv()` task.
2. In the body of `recv()` call `get_payload()` to retrieve the payload.
For now, this is the only content of the `recv()` routine. More will be added in later labs.
3. Declare the `get_payload()` task.
4. In `get_payload()`, delete content of `pkt2cmp_payload[$]`.
This is necessary to remove potential residues from previous packet.
5. Continuing in `get_payload()`, wait for the falling edge of the output frame signal.

Lab 3


Question 1. How do you implement a watchdog timer for detecting the negative edge of `rtr_io.cb.frameo_n[da]`

Hint: Remember to put it in an enclosing fork-join to prevent disable fork problems discussed in the class slides.

Also remember the caveats against using

`@(negedge rtr_io.cb.frameo_n[da])`


.....
.....
.....
.....



6. Continue to develop `get_payload()` routine by sampling the output of router:
 - Loop until end of frame is detected.
 - Within the loop, assemble a byte of data at a time (minimum of 8 clock cycles). Then, store each 8-bit datum into the `pkt2cmp_payload[$]` queue.
7. If payload is not byte aligned, print a message to terminal and end simulation.

Task 4. Develop The Checker

The last step in completing this basic testbench is to develop the checker for checking the output of the router.



1. Create a function called **compare()** which returns a single bit and has a pass-by-reference string argument.
2. In the body of **compare()**, compare the data in **payload[\$]** (reference packet data) and **pkt2cmp_payload[\$]** (sampled packet data) to verify that the payload is received correctly.
 - If the sizes of the **payload[\$]** and **pkt2cmp_payload[\$]** arrays do not match, set the string argument with a description of the error, return a value of 0 and terminate the subroutine.
 - If the data in **payload[\$]** and **pkt2cmp_payload[\$]** arrays do not match, set the string argument with a description of the error, return a value of 0 and terminate the subroutine. You can directly compare arrays in SystemVerilog using the **==** operator.
 - If the data in **payload[\$]** and **pkt2cmp_payload[\$]** arrays are the same, set the string argument with a description of the success, return a value of 1 and terminate the subroutine.
3. Create a function called **check()** which returns a void type.
4. In the body of **check()**, declare a string variable message and a counter for packets **pkts_checked**.

Lab 3

5. In the body of `check()`, call the `compare()` function to check the packet received.
 - If an error is detected, print the error message to terminal then end simulation.
 - If check is successful, print a message indicating the number of packets successfully checked to terminal.
6. Save and close the file.

Task 5. Compile and Run

1. Use `make` script to compile and run your program.
2. Make sure the simulation completes successfully. Correct all errors.

Task 6. Test All Ports

1. Modify your test program to randomly generate `sa` and `da`.
2. Extend your testbench to send **2000** packets.
3. Use `make` script to compile and run your program.
4. Make sure the simulation still completes successfully.

Task 7. Expand To Detect RTL Error

1. Run your testbench against a bad rtl code

A bad RTL code has been included in the rtl directory. The script to run your testbench is already embedded in the Makefile. To run the script, type the following command.

Compile and run simulation

```
> make bad
```

If your simulation stops on an error you are done. If not, your testbench is not working properly. Correct your testbench and try again.

Congratulations, you completed Lab 3!

Answers / Solutions

Task 3. Develop Monitor Routines

5. Continuing in `get_payload()`, wait for the falling edge of the output frame signal.

Question 2. How do you implement a watchdog timer for negative edge of `rtr_io.cb.frameo_n[da]`

There is a potential that simulation may run forever if there is an error in testbench coding.

Consider the following code:

```
for (i = 0; i < run_for_n_packets; i++)  
    gen();  
    fork  
        begin  
            send();  
            recv();  
        end  
    join
```

The `send()` and `recv()` routines were supposed to be called concurrently inside a `fork-join` construct. The coder mistakenly put both inside a `begin-end` construct making execution of `send()` and `recv()` serial rather than concurrent.

If you simply use the negative edge of `rtr_io.cb.frameo_n[da]` to detect the beginning of a new packet at the router's output the simulation would run forever, you would never know that there was a mistake or what the mistake was.

To prevent this type of runaway simulation, you can add a watchdog timer and have the routines timeout if a problem similar to the example occurs. The following code sample illustrates just such a watchdog timer implementation:

```
fork
begin: wd_timer_fork
fork: frameo_wd_timer
    begin //see class notes to understand this block
        wait(rtr_io.cb.frameo_n[da]  !== 0);
        @(rtr_io.cb iff(rtr_io.cb.frameo_n[da]  === 0));
    end
    begin
        repeat(1000) @rtr_io.cb;
        $display("\n%m\n[ERROR] %t Frame signal timed out!\n", $realtime);
        $finish;
    end
join_any: frameo_wd_timer
disable fork;
end: wd_timer_fork
join
```

test.sv Solution:

```
program automatic test(router_io.TB rtr_io);
    int run_for_n_packets;           // number of packets to test
    bit[3:0] sa;                  // source address
    bit[3:0] da;                  // destination address
    logic[7:0] payload[$];        // expected packet data array
    logic[7:0] pkt2cmp_payload[$]; // actual packet data array

initial begin
    $vcdpluson;

    run_for_n_packets = 2000;
    reset();
    repeat(run_for_n_packets) begin

        gen();

        fork
            send();
            recv();
        join

        check();

    end
    repeat(10) @rtr_io.cb;
end

task reset();
    rtr_io.reset_n = 1'b0;
    rtr_io.cb.frame_n <= '1;
    rtr_io.cb.valid_n <= '1;
    repeat(2) @rtr_io.cb;
    rtr_io.cb.reset_n <= 1'b1;
    repeat(15) @rtr_io.cb;
endtask: reset

task gen();
    sa = $urandom;
    da = $urandom;
    payload.delete(); //clear previous data
    repeat($urandom_range(4,2))
        payload.push_back($urandom);
endtask: gen

task send();
    send_addrs();
    send_pad();
    send_payload();
endtask: send
```

Continued...

```
...Continued from previous page
task send_addrs();

rtr_io.cb.frame_n[sa] <= 1'b0; //start of packet
for(int i=0; i<4; i++) begin
    rtr_io.cb.din[sa] <= da[i]; //i'th bit of da
    @rtr_io.cb;
end

endtask: send_addrs

task send_pad();

rtr_io.cb.frame_n[sa] <= 1'b0;
rtr_io.cb.din[sa] <= 1'b1;
rtr_io.cb.valid_n[sa] <= 1'b1;
repeat(5) @rtr_io.cb;

endtask: send_pad

task send_payload();

foreach(payload[index])
    for(int i=0; i<8; i++) begin
        rtr_io.cb.din[sa] <= payload[index][i];
        rtr_io.cb.valid_n[sa] <= 1'b0; //driving a valid bit
        rtr_io.cb.frame_n[sa] <= ((i == 7) && (index == (payload.size() - 1)));
        @rtr_io.cb;
    end
    rtr_io.cb.valid_n[sa] <= 1'b1;
endtask: send_payload

task recv();

//Lab 3 - Task 3, Step 2
//
//In recv() task call get_payload() to retrieve payload
//ToDo
get_payload();

endtask: recv

task get_payload();

pkt2cmp_payload.delete();

fork
begin: wd_timer_fork
fork: frameo_wd_timer
begin
    wait(rtr_io.cb.frameo_n[da] !== 0);
    @(rtr_io.cb iff(rtr_io.cb.frameo_n[da] === 0));
end
begin                                //this is another thread
repeat(1000) @rtr_io.cb;
$display("\n%m\n[ERROR] %t Frame signal timed out!\n", $realtime);
$finish;
end

```

Continued...

Lab 3

Answers / Solutions

```
...Continued from previous page
join_any: frameo_wd_timer
    disable fork;
    end: wd_timer_fork
join

forever begin
    logic[7:0] datum;
    for(int i=0; i<8; i=i)  begin //i=i prevents VCS warning messages
        if(!rtr_io.cb.valido_n[da])
            datum[i++] = rtr_io.cb.dout[da];
        if(rtr_io.cb.frameo_n[da])
            if(i==8) begin //byte aligned
                pkt2cmp_payload.push_back(datum);
                return;           //done with payload
            end else begin
                $display("\n%m\n[ERROR] %t Packet payload not byte aligned!\n",
$realtime);
                $finish;
            end
            @rtr_io.cb;
        end
        pkt2cmp_payload.push_back(datum);
    end
endtask: get_payload

function bit compare(ref string message);

    if(payload.size() != pkt2cmp_payload.size()) begin
        message = "Payload size Mismatch:\n";
        message = { message, $sformatf("payload.size() = %0d,
            pkt2cmp_payload.size() = %0d\n", payload.size(), pkt2cmp_payload.size()) };
        return (0);
    end
    if(payload == pkt2cmp_payload) ;
    else begin
        message = "Payload Content Mismatch:\n";
        message = { message, $sformatf("Packet Sent:      %p\nPkt Received:      %p",
payload, pkt2cmp_payload) };
        return (0);
    end
    message = "Successfully Compared";
    return(1);
endfunction: compare

function void check();

    string message;
    static int pkts_checked = 0;

    if (!compare(message)) begin
        $display("\n%m\n[ERROR] %t Packet #%0d %s\n", $realtime, pkts_checked,
            message);
        $finish;
    end
    $display("[NOTE] %t Packet #%0d %s", $realtime, pkts_checked++, message);
endfunction: check

endprogram: test
```

Lab 3-14

Self-Checking
Synopsys SVTB Workshop

4

OOP Encapsulation

Learning Objectives

After completing this lab, you should be able to:

- Encapsulate packet information into a **Packet** class
- Utilize **randomization** in **Packet** class to randomly generate source address, destination address and payload
- Create two **Packet** objects, one for the input into the DUT, the other for reconstructing the output of the DUT
- Use the `compare()` method embedded in the **Packet** objects to verify the correctness of DUT operation



Lab Duration:
60 minutes

Lab 4

Getting Started

In Lab 3, you added the monitor and self-check. In this lab, you will encapsulate the packet information into a class structure. You will create random Packet objects in the generator then send, receive and check the correctness of the DUT using these Packet objects.

The architecture is shown below:




Figure 1. Lab 4 testbench architecture

Lab 4-2

OOP Encapsulation
Synopsys SVTB Workshop

Lab Overview




Figure 2. Diagram of Lab Exercise

Note: You will find Answers for all questions and solutions in the Answers / Solutions at the end of this lab.

Working with Objects

Task 1. Copy Files from Lab 3's Solutions directory

1. Go into the lab4 directory.

```
> cd ../../lab4
```

2. Copy the source files in the **solutions/lab3** directory into the current directory with **make** script.

```
> make copy
```

(If you chose to use your own lab files from lab3, type “**make mycopy**”.)

Task 2. Create a Packet Class File

Create a Packet class to encapsulate the packet information.

1. Open the **Packet.sv** file in an editor.
2. Declare a class definition for Packet as follows:

```
class Packet;  
  
endclass: Packet
```

3. Use macros as a guard against multiple compilation before and after the class statements.

```
`ifndef INC_PACKET_SV  
`define INC_PACKET_SV  
class Packet;  
  
endclass: Packet  
`endif
```

4. In the body of the **Packet** class, create the following properties:
 - **rand bit[3:0] sa, da;** // random port selection
 - **rand logic[7:0] payload[\$];** // random payload array
 - **string name;** // (see description below)

Individual test **Packet** objects will be created by the **gen()** routine. For these **Packet** objects, you will want to tag each one uniquely to identify the object.

The string property **name** is the unique identifier. Displaying the **name** property as part of the printed message will be very helpful during the debugging process.

Task 3. Define Packet Property Constraints

1. Immediately after the property declarations, add a constraints block to limit **sa** and **da** to **0:15**, and the **payload.size()** to **2:4**.

Task 4. Define Packet Class Method Prototypes

1. Add the following method prototype declarations in the body of **Packet** class:

```
class Packet;
...
extern function new(string name = "Packet");
extern function bit compare(Packet pkt2cmp, ref string message);
extern function void display(string prefix = "NOTE");
extern function Packet copy();
endclass: Packet
```

Task 5. Define Packet Class **new()** Constructor

The class constructor **new()** is used to initialize object properties. For **Packet** objects, most properties will be set with a call to **randomize()**. The one property that needs to be initialized in the constructor is **name**.

1. Outside of the class body, create the constructor **new()** method. Make sure to reference the method back to the class via the **Packet::** notation.
2. Inside the constructor body, assign the class property **name** with the string passed in via the argument:

```
function Packet::new(string name);
    this.name = name;
endfunction: new
```

This mechanism makes the string passed in through the constructor argument accessible to all other methods of the **Packet** class.

Lab 4

Task 6. Define Packet compare() Method

For self checking, comparing contents of two data objects is a common need. It is a good idea to build the compare method within the data object.

1. Cut and paste `compare()` from `test.sv` into the `Packet` class at the end of the file.

Note: An alternative to cut and paste is to concatenate (cat) content of `test.sv` into `Packet.sv` then keep only `compare()` in `Packet.sv`.

2. Reference the method to `Packet` class with `::` notation.
3. Modify the **argument list** to contain a `Packet` handle:

```
function bit Packet::compare(Packet pkt2cmp, ref string message);
```

4. Inside the `compare()` method, change `pkt2cmp_payload` to reference the class property `pkt2cmp.payload`.

Task 7. Define Packet display() Method

It is helpful during the debugging process to print out the contents of a packet. To ease this effort, a display method should be defined for the Packet class to print the content of the Packet object to the console.

1. Outside of the class body, create the `display()` method:

```
function void Packet::display(string prefix = "NOTE");
```

2. Inside the method body, print a formated content of the object to terminal.

In the printed message, you should also include the string passed via the argument list. This string can be set by user to differentiate between different types of message: ERROR, WARNING, DEBUG, etc.

Or, you can just use the display routine saved in the `.display` file in solutions directory (``endif` must be after the inserted code):


```
> cat ../../solutions/lab4/.display >> Packet.sv
```

3. The `copy()` method is already written for you.
4. Save and close the file.

Task 8. Modify test.sv to use Packet class

The **Packet** class now encapsulates the router packet information.

You will now generate random **Packet** objects, then send and receive packets through the router based on these random **Packet** objects.



The following steps take you through this process.

1. Open **test.sv** in an editor.
2. Inside the program block, add an include statement for the **Packet** class file.
3. Create and construct two program global **Packet** objects **pkt2send** and **pkt2cmp**.

Task 9. Modify gen() Task To Generate Packet objects

You will now use the **Packet** objects just created to generate random stimulus.

1. In the **gen()** task, delete all existing code.
2. Declare a static int **pkts_generated** variable (initialize to 0) to keep track of how many packets were generated by the generator
3. Set the **name** property of **pkt2send** to a unique string (use the **pkts_generated** variable value as part of the string)
4. Randomize the **Packet** object **pkt2send** (print an error message to terminal and end simulation if randomization fails)
5. Update all program global variables, **sa**, **da** and **payload** with values from the randomized **pkt2send** object.

Lab 4

This makes the content of **pkt2send** object visible to all components of the program. These program global variables are only an interim solution to simplify development of subroutines in the program block. In the next lab, these variables will all be deleted and migrated inside individual testbench component objects.

When done, your code might look something like the following:

```
task gen();
    static int pkts_generated = 0;
    pkt2send.name = $sformatf("Packet [%0d]", pkts_generated++);
    if (!pkt2send.randomize()) begin
        $display("\n%m\n[ERROR] %t Randomization Failed!", $realtime);
        $finish;
    end
    sa = pkt2send.sa;
    da = pkt2send.da;
    payload = pkt2send.payload;
endtask: gen
```

Task 10. Modify **recv()** task

In the **recv()** task, the payload sampled from the output of the router needs to be assembled into a Packet object (**pkt2cmp**). This Packet object will then be used in the checking process against the **pkt2send** object.

In **recv()** task:

*Before calling the **get_payload()** task do the following:*

1. Create a static int variable **pkt_cnt** to track the number of packets received (should be initialized to 0).

*After the **get_payload()** task do the following:*

2. Assign **pkt2cmp.da** with the value of program global variable **da**.
3. Assign **pkt2cmp.payload** with the values from **pkt2cmp_payload** array.
4. Set a unique **name** for the **pkt2cmp** object. (use **pkt_cnt** value as part of the string).

When finished, the **recv()** task should look like:

```
task recv();
    static int pkt_cnt = 0;
    get_payload();
    pkt2cmp.da = da;
    pkt2cmp.payload = pkt2cmp_payload;
    pkt2cmp.name = $sformatf("rcvdPkt [%0d]", pkt_cnt++);
endtask: recv
```

Task 11. Modify the `check()` function

In the **check()** function, you will use the **compare()** method built into the **Packet** object to verify the content of the **Packet** object sent and received. You should use the **display()** method in the **Packet** object to assist in debugging errors.

1. Replace the **compare()** call with a call to the **compare()** method within the **Packet** object. (The two objects you want to compare are the program global objects **pkt2send** and **pkt2cmp**).
2. Make use of the **display()** method when an error is detected.

When finished, the **check()** routine should look like:

```
function void check();
    string message;
    static int pkts_checked = 0;
    if (!pkt2send.compare(pkt2cmp, message)) begin
        $display("\n%m\n[ERROR] %t Packet#%0d %s\n", $realtime, pkts_checked,
message);
        pkt2send.display("ERROR");
        pkt2cmp.display("ERROR");
        $finish;
    end
    $display("[NOTE] %t Packet#%0d %s", $realtime, pkts_checked++, message);
endfunction: check
```

Lab 4

Task 12. Check and Save file

1. Make sure you have deleted the `compare()` routine in `test.sv`.
2. Save and close the file.

Task 13. Compile and Run

1. Use `make` script to compile and run your program.

> `make`

Debug any error you find.

2. If the testbench runs successfully, execute the following script which runs the testbench on a bad RTL code.

> `make bad`

If the simulation finds an error, you are done.

If the simulation does not find an error, you have a problem with your testbench. You must debug the error in your testbench.

Congratulations, you completed Lab 4!

Answers / Solutions

test.sv Solution:

```

program automatic test(router_io.TB rtr_io);
    // The following program variables will be seen by the included files without
extern
    int run_for_n_packets;      // number of packets to test
    `include "Packet.sv"

    // The following program variables can be seen by the included files with
extern
    bit[3:0] sa;                // source address
    bit[3:0] da;                // destination address
    logic[7:0] payload[$];      // expected packet data array
    logic[7:0] pkt2cmp_payload[$]; // actual packet data array
    Packet    pkt2send = new();   // expected Packet object
    Packet    pkt2cmp = new();    // actual Packet object

initial begin
    $vcapluson;
    run_for_n_packets = 2000;
    reset();
    repeat(run_for_n_packets) begin
        gen();
        fork
            send();
            recv();
        join
        check();
    end
    repeat(10) @rtr_io.cb;
end

task reset();
    rtr_io.reset_n <= 1'b0;
    rtr_io.cb.frame_n <= '1;
    rtr_io.cb.valid_n <= '1;
    repeat(2) @rtr_io.cb;
    rtr_io.cb.reset_n <= 1'b1;
    repeat(15) @rtr_io.cb;
endtask: reset

task gen();
    static int pkts_generated = 0;
    pkt2send.name = $sformatf("Packet[%0d]", pkts_generated++);
    if (!pkt2send.randomize()) begin
        $display("\n%m\n[ERROR] %t gen(): Randomization Failed!", $realtime);
        $finish;
    end
    sa = pkt2send.sa;
    da = pkt2send.da;
    payload = pkt2send.payload;
endtask: gen

```

Continued...

```
...Continued from previous page
task send();
    send_addrs();
    send_pad();
    send_payload();
endtask: send

task send_addrs();
    rtr_io.cb.frame_n[sa] <= 1'b0;
    for(int i=0; i<4; i++) begin
        rtr_io.cb.din[sa] <= da[i];
        @rtr_io.cb;
    end
endtask: send_addrs

task send_pad();
    rtr_io.cb.frame_n[sa] <= 1'b0;
    rtr_io.cb.valid_n[sa] <= 1'b1;
    rtr_io.cb.din[sa] <= 1'b1;
    repeat(5) @rtr_io.cb;
endtask: send_pad

task send_payload();
    foreach(payload[index]) begin
        for(int i=0; i<8; i++) begin
            rtr_io.cb.din[sa] <= payload[index][i];
            rtr_io.cb.valid_n[sa] <= 1'b0;
            rtr_io.cb.frame_n[sa] <= (index == (payload.size() - 1)) && (i == 7);
            @rtr_io.cb;
        end
    end
    rtr_io.cb.valid_n[sa] <= 1'b1;
endtask: send_payload

task recv();
    static int pkt_cnt = 0;
    get_payload();
    pkt2cmp.da = da;
    pkt2cmp.payload = pkt2cmp_payload;
    pkt2cmp.name = $sformatf("rcvdPkt[%0d]", pkt_cnt++);
endtask: recv
```

Continued...

Lab 4-12

OOP Encapsulation
Synopsys SVTB Workshop

```
...Continued from previous page
task get_payload();
    pkt2cmp_payload.delete();
    fork
        begin: wd_timer_fork
        fork: frameo_wd_timer
            begin //see class notes to understand this block
                wait(rtr_io.cb.frameo_n[da] != 0);
                @(rtr_io.cb iff(rtr_io.cb.frameo_n[da] == 0));
            end
            begin
                repeat(1000) @rtr_io.cb;
                $display("\n%m\n[ERROR] %t Frame signal timed out!\n", $realtime);
                $finish;
            end
        join_any: frameo_wd_timer
        disable fork;
        end: wd_timer_fork
    join
    forever begin
        logic[7:0] datum;
        for (int i=0; i<8; ) begin
            if (!rtr_io.cb.valido_n[da])
                datum[i++] = rtr_io.cb.dout[da];
            if (rtr_io.cb.frameo_n[da])
                if (i == 8) begin
                    pkt2cmp_payload.push_back(datum);
                    return;
                end
                else begin
                    $display("\n%m\n[ERROR] %t Payload not byte aligned!\n", $realtime);
                    $finish;
                end
            @rtr_io.cb;
        end
        pkt2cmp_payload.push_back(datum);
    end
endtask: get_payload

function void check();
    string message;
    static int pkts_checked = 0;
    if (!pkt2send.compare(pkt2cmp, message)) begin
        $display("\n%m\n[ERROR] %t Packet #%-0d %s\n", $realtime, pkts_checked,
message);
        pkt2send.display("ERROR");
        pkt2cmp.display("ERROR");
        $finish;
    end
    $display("[NOTE] %t Packet #%-0d %s", $realtime, pkts_checked++, message);
endfunction: check

endprogram: test
```

Packet.sv Solution:

```

`ifndef INC_PACKET_SV
`define INC_PACKET_SV
class Packet;
    rand bit[3:0] sa, da;           //random port selection
    rand logic[7:0] payload[$];    //random payload array
        string name;               //unique identifier
    constraint Limit {
        sa inside {[0:15]};
        da inside {[0:15]};
        payload.size() inside {[2:4]};
    }

    extern function new(string name = "Packet");
    extern function bit compare(Packet pkt2cmp, ref string message);
    extern function void display(string prefix = "NOTE");
    extern function Packet copy();
endclass: Packet

function Packet::new(string name);
    this.name = name;
endfunction: new

function bit Packet::compare(Packet pkt2cmp, ref string message);
    if (payload.size() != pkt2cmp.payload.size()) begin
        message = "Payload Size Mismatch:\n";
        message = { message, $sformatf("payload.size() = %0d",
pkt2cmp.payload.size() = %0d\n", payload.size(), pkt2cmp.payload.size()) };
        return(0);
    end
    if (payload == pkt2cmp.payload) ;
    else begin
        message = "Payload Content Mismatch:\n";
        message = { message, $sformatf("Packet Sent: %p\nPkt Received: %p",
payload, pkt2cmp.payload) };
        return(0);
    end
    message = "Successfully Compared";
    return(1);
endfunction: compare

function void Packet::display(string prefix);
    $display("[%s]@%t %s sa = %0d, da = %0d", prefix, $realtime, name, sa, da);
    foreach(payload[i])
        $display("[%s]@%t %s payload[%0d] = %0d", prefix, $realtime, name, i,
payload[i]);
endfunction: display

function Packet Packet::copy();
    Packet pkt_copy = new();
    pkt_copy.sa = this.sa;
    pkt_copy.da = this.da;
    pkt_copy.payload = this.payload;
    return(pkt_copy);
endfunction

`endif

```

5

Broad Spectrum Verification

Learning Objectives

After completing this lab, you should be able to:

- Build a Generator transactor class
- Build a Driver class
- Build a Receiver class
- Build a Scoreboard class
- Expand the testbench to drive and monitor all input and output ports concurrently



Lab Duration:
90 minutes

Lab 5

Getting Started

In lab 4, you created an encapsulated packet. But, because you had only one driver and monitor, you were only able to drive a single input and output port at a time.

In this lab, you will encapsulate the generator, driver, monitor and check routines into **Generator** class, **Driver** class, **Receiver** class and **Scoreboard** class respectively. You will then build a testbench architecture that is capable of exercising all ports simultaneously.

To facilitate passing of **Packet** object from transactor to transactor, you will use the SystemVerilog built-in **mailbox** class as the communication mechanism.

The resulting architecture is shown below:




Figure 1. Lab 5 Encapsulate transactors for broad-spectrum verification




Figure 2. Lab 5 testbench architecture

Lab 5

Lab Overview




Figure 3. Diagram of Lab Exercise

Note:

You will find Answers for all questions and solutions in the Answers / Solutions at the end of this lab.

Broad-Spectrum Verification

Task 1. Copy Files from Lab 4's solutions directory

1. Go into the lab5 directory.

```
> cd ../lab5
```

2. Copy the source files in the **solutions/lab4** directory into the current directory with the **make** script.

```
> make copy
```

Task 2. Develop Driver class

A **DriverBase** class which encapsulates the driver routines and the program global variables used in previous labs is already developed for you. You will extend from this base class to implement a new **Driver** class.

1. Open the existing **DriverBase.sv** file in an editor
2. Examine the **DriverBase** class.

In this base class, the following properties are declared:

- **virtual router_io.TB rtr_io;** // interface signal
- **string name;** // unique identifier
- **bit[3:0] sa, da;** // source and destination addresses
- **logic[7:0] payload[\$];** // Packet payload
- **Packet pkt2send;** // stimulus Packet object

The **rtr_io** property defines the interface signals for the Drivers to drive. The **name** property uniquely identifies the object. Both properties will be set by the constructor **new()**.

The **sa**, **da**, **payload** and **pkt2send** properties are the program global variables that you had used in previous labs. You will set these properties in the **Driver** class to be developed in the next few steps. Since these are class properties, all methods in the class can access these properties directly.

For each of the methods in the class, there is an **if-\$display()** combination at the beginning of the subroutine. It is often helpful during debugging to be able to see a trace of the subroutine execution sequences. This **if-\$display()** combination will let you control whether or not to print subroutine sequence tracing to the terminal. The variable **TRACE_ON** is a program global variable that you will declare and control later in the **test.sv** file.

3. Close the file.

Lab 5

4. Open the existing **Driver.sv** file in an editor.

The **Driver** class is derived from the **DriverBase** class. Three properties and two method prototypes are declared for you.

The **in_box** will be used to pass **Packet** objects from the Generator to the Driver. The **out_box** will be used to pass **Packet** objects from the Driver to the Scoreboard. The **in_box** and **out_box** are of type **pkt_mbox**. This is a typed **mailbox** defined in the file **router_test.h** as shown below.

```
typedef class Packet;
typedef mailbox #(Packet) pkt_mbox;
```

The **sem[]** array will be used as an arbitration mechanism for preventing multiple input ports trying to drive the same output port at the same time.

The constructor method has five variables in the argument list. The **name** argument allows user to assign a unique identifier. The **port_id** argument sets the Driver to drive a specific input port. The **sem[]** argument is a set of semaphore bins for the Driver to self-arbitrate access to output ports. The **in_box** argument is a mailbox, which connects the Driver to the Generator. The **out_box** argument is a mailbox which connects the Driver to the Scoreboard.

The **start()** method retrieves **Packet** object from **in_box**. Within the method, drive the packet through the DUT with a call to **send()**, if the selected destination address port is not already in use by another driver.

```
`ifndef INC_DRIVER_SV
`define INC_DRIVER_SV
`include "DriverBase.sv"
class Driver extends DriverBase;
    pkt_mbox in_box;                      // Generator mailbox
    pkt_mbox out_box;                     // Scoreboard mailbox
    semaphore sem[];                     // output port arbitration
    extern function new(...);
    extern virtual task start();
endclass

function Driver::new(string name, int port_id, semaphore sem[],
    pkt_mbox in_box, out_box, virtual router_io.TB rtr_io);
endfunction

task Driver::start();
endtask: start
`endif
```

Task 3. Fill in Driver class new() method

1. In the body of the externally declared constructor **new()**, call **super.new()** with the **name** and **rtr_io** arguments.
2. Add a tracing statement after the call to **super.new()** as follows:

```
if (TRACE_ON) $display("[TRACE] %t %s:%m", $realtime, this.name);
```

3. Assign the class property **sa** (defined in base class) to the value passed in via **port_id**.
4. Complete the constructor development by assigning class properties **sem[]**, **in_box** and **out_box** with the values passed in via the argument list.

Task 4. Fill in Driver Class start() Method

Each transactor object you instantiate in the test program will need a mechanism to start operation. You will standardize the name of this method to be **start()**.

For the **Driver**, the **start()** method will execute an infinite loop. In each iteration of the loop, a **Packet** object will be retrieved from **in_box**. This **Packet** object content will then be driven through the DUT via a call to **send()**. Once the **Packet** object processing is completed, the **Packet** object is passed on to **Scoreboard** via the **out_box**.

Since the **Driver** object, when started, is expected to run concurrently with all other components of the testbench, all contents of the **start()** method with the exception of the trace statement must be inside a non-blocking **fork-join** construct.

1. In the existing **start()** method body, add a trace statement.
2. After the trace statement, create a **non-blocking fork-join** block.
3. Inside the **fork-join** construct, create a single infinite loop.
4. Each iteration through the loop do the following:
 - a) Retrieve a **Packet** object (**pkt2send**) from **in_box**.
 - b) If the **sa** property in the retrieved **Packet** object does not match **this.sa**, continue on to the next iteration of the loop.
 - c) If the retrieved **Packet** **sa** does match **this.sa**, update the **da** and **payload** class properties with the content of **pkt2send**.
 - d) Use the semaphore **sem[]** array to arbitrate for access to the output port specified by **da**.

Lab 5

- e) Once the arbitration is successful, call **send()** to drive the packet through the DUT.
 - f) When **send()** completes, deposit **Packet** object into **out_box**.
 - g) Put the semaphore key back into its bin in the final step of the loop.
5. Save and close the file.

Task 5. Develop Receiver Class

1. Open the existing **Receiver.sv** skeleton file in an editor.

```
'ifndef INC_RECEIVER_SV
`define INC_RECEIVER_SV
`include "ReceiverBase.sv"
class Receiver extends ReceiverBase;
    pkt_mbox out_box; // Scoreboard mailbox
    extern function new(...);
    extern virtual task start();
endclass
function Receiver::new(string name, int port_id, pkt_mbox
out_box);
endfunction
task Receiver::start();
endtask: start
`endif
```

Task 6. Fill in Receiver Class new()

1. In the body of the externally declared constructor **new()**, call **super.new()** with the **name** and **rtr_io** argument.
2. Add a tracing statement after the call to **super.new()**.
3. Assign the class property **da** to the value passed in via **port_id**.
4. Assign class property **out_box** with values passed in via the argument list.

Task 7. Fill in Receiver Class start() Method

The **start()** method will execute a non-blocking infinite loop. In each iteration of the infinite loop, reconstruct a **Packet** object (**pkt2cmp**) monitored from DUT. Once, retrieved, this **Packet** object will be passed to Scoreboard via an out mailbox.

1. In the **start()** method body, add a trace statement.
2. After the trace statement, create a **non-blocking** concurrent process thread.
3. Inside the **fork-join** construct, create a single infinite loop code block.
4. Each iteration through the loop do the following:
 - a) Call **recv()** to retrieve a **Packet** object from DUT
 - b) Deposit a copy of the **Packet** object (**pkt2cmp**) retrieved from DUT into **out_box**. Use the **copy()** method of the **Packet** class.
5. Save and close the file.

Task 8. Examine the Generator class

In the interest of saving time during lab, a **Generator.sv** file is written for you. It encapsulates the **gen()** routine that you had completed earlier and includes a **start()** method similar to what you have done for the Drivers and Receivers.

There are two significant differences in the **start()** method you should know about. First, the **start()** method loop is controlled by the program global **run_for_n_packets** variable. If **run_for_n_packets** is ≤ 0 , then the loop will be infinite. If it is > 0 , then, the loop will stop after **run_for_n_packets** iterations. Second, after **gen()** method is called, a copy of the randomized **Packet** object (**pkt2send**) is created and sent to the **Drivers** via **out_box** mailboxes. Note that the **gen()** routine has a static variable **pkts_generated** to count the packets generated across all generators.

The Generator generates packets for the particular driver it is connected to, by constraining packets to source address values matching its **port_id** variable.

Examine the content of the **Generator.sv** file if you are interested.

Lab 5

Task 9. Examine The Scoreboard Class

A **Scoreboard.sv** file has also been written for you. It is mainly an encapsulation of the **check()** routine you have already written.

The main new feature is the implementation of mailboxes to allow communication between the **Scoreboard** and the **Drivers** and **Receivers**.

A **Driver** will deposit the **Packet** objects it has just sent to the DUT into the **driver_mbox** mailbox. A **Receiver** will deposit the **Packet** object it has just retrieved from the DUT into the **receiver_mbox** mailbox.


When the Scoreboard finds a **Packet** object in the **receiver_mbox**, it will first save this object handle as **pkt2cmp**. Then, it will push all **Packet** objects found in the **driver_mbox** onto a **refPkt[\$]** queue. Afterwards, on the basis of the output port address (**da**) in **pkt2cmp** object, it will try to locate the corresponding reference **Packet** in **refPkt[\$]** and compare the content. If no corresponding reference **Packet** is found, an error is reported.

When the number of **Packet** objects checked matches the global variable **run_for_n_packets**, an event flag called **DONE** is triggered. This **DONE** flag will allow the simulation to terminate gracefully at the appropriate time.

Examine the content of the **Scoreboard.sv** file if you are interested.

Task 10. Modify `test.sv` To Use These New Classes

You will now modify the testbench to have 16 generators, one scoreboard, 16 drivers and 16 receivers.



1. Open `test.sv` in an editor.
2. Delete all program global variables except `run_for_n_packets`.
3. Create an int variable `TRACE_ON` and initialize it to 0 (change to 1 if subroutine execution tracing is desired for debugging).

Lab 5

4. Following the two program global variables, add include statements for all header files and the new class files (**router_test.h**, **Driver.sv**, **Receiver.sv**, **Generator.sv**, **Scoreboard.sv**).
5. Following the include statements, add the following program global variables:
 - **semaphore sem[];** // prevent output port collision
 - **Driver drvr[];** // driver objects
 - **Receiver rcvr[];** // receiver objects
 - **Generator gen[];** // generator objects
 - **Scoreboard sb;** // scoreboard object
6. Delete all content of the initial block except for one:

```
initial begin
    run_for_n_packets = 2000;
end
```

Add the following:

7. Construct all objects declared in the program block. Make sure the mailboxes are connected correctly:
Each Driver to corresponding Generator (**gen[i].out_box**)
All Drivers to one Scoreboard mailbox (**sb.driver_mbox**)
All Receivers to one Scoreboard mailbox (**sb.receiver_mbox**)
8. After all objects are constructed, call **reset()** to reset the DUT.
9. Then, start all transactors (Generators, Scoreboard, Drivers and Receivers).
10. Finally, before the end of the program, block until Scoreboard's **DONE** event flag is set.

Task 11. Misc checks

1. The following subroutines should be deleted from **test.sv**.

- **gen()**
- **send()**
- **send_addr()**
- **send_pad()**
- **send_payload()**
- **recv()**
- **get_payload()**
- **check()**

(Make sure **reset()** is NOT deleted)

When done, your **test.sv** file should look like:

```

program automatic test(router_io.TB rtr_io);
    int run_for_n_packets; // number of packets to test
    int TRACE_ON = 0; // subroutine tracing control

    `include "router_test.h"
    `include "Packet.sv"
    `include "Driver.sv"
    `include "Receiver.sv"
    `include "Generator.sv"
    `include "Scoreboard.sv"

    semaphore sem[]; // prevent output port collision
    Driver drvr[]; // driver objects
    Receiver rcvr[]; // receiver objects
    Generator gen[]; // generator objects
    Scoreboard sb; // scoreboard object

    initial begin
        $vcapluson;
        run_for_n_packets = 2000;
        sem = new[16];
        drvr = new[16];
        rcvr = new[16];
        gen = new[16];
        sb = new("sb");
        foreach (sem[i]) sem[i] = new(1);
        foreach (gen[i]) gen[i] = new($sformatf("gen[%0d]", i), i);
        foreach (drvr[i])
            drvr[i] = new($sformatf("drvr[%0d]", i), i, sem,
            gen[i].out_box, sb.driver_mbox, rtr_io);
        foreach (rcvr[i])
            rcvr[i] = new($sformatf("rcvr[%0d]", i), i,
            sb.receiver_mbox, rtr_io);
        reset();
        sb.start();
        foreach (gen[i]) gen[i].start();
        foreach (drvr[i]) drvr[i].start();
        foreach (rcvr[i]) rcvr[i].start();
        wait (sb.DONE.triggered);
    end
    task reset();
        if (TRACE_ON) $display("[TRACE] %t %m", $realtime);
        rtr_io.reset_n <= 1'b0;
        rtr_io.cb.frame_n <= '1;
        rtr_io.cb.valid_n <= '1;
        repeat(2) @(rtr_io.cb);
        rtr_io.cb.reset_n <= 1'b1;
        repeat(15) @(rtr_io.cb);
    endtask: reset
endprogram: test

```

Lab 5

2. Save and close the file.

Task 12. Compile and Run

1. Use make script to compile and run your program.

> **make**

Debug any error you find.

2. If the testbench runs successfully again, execute the following script which runs the testbench on a bad RTL code.

> **make bad**

If the simulation finds an error, you are done.

Congratulations, you completed Lab 5!

Answers / Solutions

test.sv Solution:

```

program automatic test(router_io.TB rtr_io);
    // following program variables will be seen by the included files without extern
    int run_for_n_packets;      // number of packets to test
    int TRACE_ON = 0;           // subroutine tracing control

    `include "router_test.h"
    `include "Packet.sv"
    `include "Driver.sv"
    `include "Receiver.sv"
    `include "Generator.sv"
    `include "Scoreboard.sv"

    // The following program variables can be seen by the included files withextern
    semaphore sem[];           // prevent output port collision
    Driver drvr[];             // driver objects
    Receiver rcvr[];           // receiver objects
    Generator gen[];           // generator object
    Scoreboard sb;              // scoreboard object

initial begin
    $vcappluson;
    run_for_n_packets = 2000;
    sem = new[16];
    drvr = new[16];
    rcvr = new[16];
    gen = new[16];
    sb = new("sb");
    foreach (sem[i]) sem[i] = new(1);
    foreach (gen[i]) gen[i] = new($sformatf("gen[%0d]", i, i));
    foreach (drvr[i])
        drvr[i] = new($sformatf("drvr[%0d]", i), i, sem, gen[i].out_box, sb.driver_mbox,
rtr_io);
    foreach (rcvr[i])
        rcvr[i] = new($sformatf("rcvr[%0d]", i), i, sb.receiver_mbox, rtr_io);
    reset();
    sb.start();
    foreach(gen[i]) gen[i].start();
    foreach(drvr[i]) drvr[i].start();
    foreach(rcvr[i]) rcvr[i].start();
    wait(sb.DONE.triggered);
end

task reset();
    if (TRACE_ON) $display("[TRACE] %t %m", $realtime);
    rtr_io.reset_n <= 1'b0;
    rtr_io.cb.frame_n <= '1;
    rtr_io.cb.valid_n <= '1;
    repeat(2) @rtr_io.cb;
    rtr_io.cb.reset_n <= 1'b1;
    repeat(15) @(rtr_io.cb);
endtask: reset
endprogram: test

```

Driver.sv Solution:

```
`ifndef INC_DRIVER_SV
`define INC_DRIVER_SV
`include "DriverBase.sv"
class Driver extends DriverBase;
    pkt_mbox in_box; // Generator mailbox
    pkt_mbox out_box; // Scoreboard mailbox
    semaphore sem[]; // output port arbitration

    extern function new(string name = "Driver", int port_id, semaphore
sem[], pkt_mbox in_box, out_box, virtual router_io.TB rtr_io);
    extern virtual task start();
endclass: Driver

function Driver::new(string name, int port_id, semaphore sem[],
pkt_mbox in_box, out_box, virtual router_io.TB rtr_io);
    super.new(name, rtr_io);
    if (TRACE_ON) $display("[TRACE] %t %s:%m", $realtime, this.name);
    this.sa = port_id;
    this.sem = sem;
    this.in_box = in_box;
    this.out_box = out_box;
endfunction: new

task Driver::start();
    if (TRACE_ON) $display("[TRACE] %t %s:%m", $realtime, this.name);
    fork
        forever begin
            this.in_box.get(this(pkt2send));
            if (this(pkt2send).sa != this.sa) continue;
            this.da = this(pkt2send).da;
            this.payload = this(pkt2send).payload;
            this.sem[this.da].get(1);
            this.send();
            this.out_box.put(this(pkt2send));
            this.sem[this.da].put(1);
        end
    join_none
endtask: start
`endif
```

Receiver.sv Solution:

```
`ifndef INC_RECEIVER_SV
`define INC_RECEIVER_SV
`include "ReceiverBase.sv"
class Receiver extends ReceiverBase;
    pkt_mbox out_box; // Scoreboard mailbox

    extern function new(string name = "Receiver", int port_id,
    pkt_mbox
        out_box, virtual router_io.TB rtr_io);
    extern virtual task start();
endclass: Receiver

function Receiver::new(string name, int port_id, pkt_mbox out_box,
virtual router_io.TB rtr_io);
    super.new(name, rtr_io);
    if (TRACE_ON) $display("[TRACE] %t %s:%m", $realtime, this.name);
    this.da = port_id;
    this.out_box = out_box;
endfunction: new

task Receiver::start();
    if (TRACE_ON) $display("[TRACE] %t %s:%m", $realtime, name);
    fork
        forever begin
            this.recv();
            begin
                Packet pkt = this.pkt2cmp.copy();
                this.out_box.put(pkt);
            end
        end
        join_none
    endtask: start
`endif
```

Generator.sv Solution:

```

`ifndef INC_GENERATOR_SV
`define INC_GENERATOR_SV
class Generator;
    string name;      // unique identifier
    Packet pkt2send; // stimulus Packet object
    mbox out_box;   // mailbox to Drivers
    int port_id = -1; // port_id of connected Driver
    static int pkts_generated = 0; //packet count across all generators

    extern function new(string name = "Generator", int port_id);
    extern virtual task gen();
    extern virtual task start();
endclass: Generator

function Generator::new(string name, int port_id);
    if (TRACE_ON) $display("[TRACE] %t %s:%m", $realtime, name);
    this.name = name;
    this(pkt2send) = new();
    this.out_box = new(1); //1-deep mailbox
    this.port_id = port_id;
endfunction: new

task Generator::gen();
    if (TRACE_ON) $display("[TRACE] %t %s:%m", $realtime, this.name);
    this(pkt2send).name = $sformatf("Packet[%0d]", this.pkts_generated++);
    if (!this(pkt2send).randomize()
        with {if (port_id != -1) sa == port_id;})
    begin
        $display("\n%m\n[ERROR] %t Randomization Failed!\n", $realtime);
        $finish;
    end
endtask: gen

task Generator::start();
    if (TRACE_ON) $display("[TRACE] %t %s:%m", $realtime, this.name);
    fork
        while (this.pkts_generated < run_for_n_packets ||
run_for_n_packets <= 0) begin
            this.gen();
            begin
                Packet pkt = this(pkt2send).copy();
                this.out_box.put(pkt);
            end
        end
        join_none
    endtask: start
`endif

```

Scoreboard.sv Solution:

```
`ifndef INC_SCOREBOARD_SV
`define INC_SCOREBOARD_SV
class Scoreboard;
    string name;           // unique identifier
    event DONE;           // flag to indicate goal reached
    Packet refPkt[$];    // reference Packet array
    Packet pkt2send;     // Packet object from Drivers
    Packet pkt2cmp;      // Packet object from Receivers
    pkt_mbox driver_mbox; // mailbox for Packet objects from Drivers
    pkt_mbox receiver_mbox; // mailbox for Packet objects from
Receivers

extern function new(string name = "Scoreboard",
                    pkt_mbox driver_mbox = null,
                    receiver_mbox = null);
extern virtual task start();
extern virtual function void check();
endclass: Scoreboard

function Scoreboard::new(string name, pkt_mbox driver_mbox,
receiver_mbox);
    if (TRACE_ON) $display("[TRACE] %t %s:%m", $realtime, name);
    this.name = name;
    if (driver_mbox == null) driver_mbox = new();
    this.driver_mbox = driver_mbox;
    if (receiver_mbox == null) receiver_mbox = new();
    this.receiver_mbox = receiver_mbox;
endfunction: new

task Scoreboard::start();
    if (TRACE_ON) $display("[TRACE] %t %s:%m", $realtime, this.name);
    fork
        forever begin
            this.receiver_mbox.get(this(pkt2cmp));
            while (this.driver_mbox.num()) begin
                Packet pkt;
                this.driver_mbox.get(pkt);
                this.refPkt.push_back(pkt);
            end
            this.check();
        end
    join_none
endtask: start
```

Continued...

```
...Continued from previous page
function void Scoreboard::check();
    int      index[$];
    string   message;
    static int pkts_checked = 0;
    if (TRACE_ON) $display("[TRACE] %t %s:%m", $realtime, this.name);
    index = this.refPkt.find_first_index() with (item.da ==
                                                this(pkt2cmp).da);
    if (index.size() <= 0) begin
        $display("\n%m\n[ERROR] %t %s not found in Reference Queue\n",
                 $realtime, this(pkt2cmp).name);
        this(pkt2cmp).display("ERROR");
        $finish;
    end
    this(pkt2send) = this.refPkt[index[0]];
    this.refPkt.delete(index[0]);
    if (!this(pkt2send).compare(this(pkt2cmp), message)) begin
        $display("\n%m\n[ERROR] %t Packet #%0d %s\n",
                 $realtime, pkts_checked, message);
        this(pkt2send).display("ERROR");
        this(pkt2cmp).display("ERROR");
        $finish;
    end
    $display("[NOTE] %t Packet #%0d %s", $realtime, pkts_checked++, message);
    if (pkts_checked >= run_for_n_packets)
        ->this.DONE;
endfunction: check
`endif
```

6

Functional Coverage, Using Packages, Standardizing Environments

Learning Objectives

After completing this lab, you should be able to:

- Implement Functional Coverage to determine when you are done with simulation
- Define Packages for reuse
- Import Packages for use in a test
- Understand the use of Environments and Standardized Test Methodologies



Lab Duration:
45 minutes

Lab 6

Getting Started

The question that Lab 5 did not answer is - how many packets should be sent through the router in order to test all combinations of input and output ports? With the current random stimulus based code, the answer cannot be determined. You will need to implement functional coverage.

In the first part of this lab, you will add the functional coverage components in the scoreboard class. Within the scoreboard class, you will implement functional coverage to measure the progress of your testbench and end the simulation when the testbench has fully exercised all input and output port combinations.

In Lab 5, you expanded your testbench to do broad-spectrum verification. You constructed all the components like drivers, receivers etc., and then “started” them to run the simulation. Then you coordinated the end-of-simulation. This methodology is very common to almost all testbenches that perform simulation-based verification. Can we standardize some of the components of the structure and tasks of running the simulation? This is what most standard methodologies like VMM, UVM etc. do.

In the second part of this lab you will use an “Environment” that encapsulates the components and “runs” them in a standardized manner. You will also define a package that allows reuse of standard components and class libraries.




Figure 1. Lab 6 Standardized Environments and Functional Coverage

Lab 6-2

Environments, Packages, Functional Coverage
Synopsys SVTB Workshop

Lab Overview




Figure 2. Diagram of Lab 6 Exercise

Note:

You will find Answers for all questions and solutions in the Answers / Solutions at the end of this lab.

Lab 6

Part 1: Functional Coverage

Task 1. Copy Files from Lab 5's Solutions directory

1. Go into the lab6 directory.

```
> cd ../lab6a
```

2. Copy the source files in the **solutions/lab5** directory into the current directory with **make** script.

```
> make copy
```

(If you chose to use your own lab files from Lab5, type “**make mycopy**”.)

Task 2. Create a covergroup in Scoreboard Class

First thing in implementing functional coverage in SystemVerilog is to define the coverage groups. Within the coverage groups, we can define coverage bins, bin update event timing and coverage goal.

Coverage bins should be created for each input port and output port. Then, cross coverage bins for all combinations of the input and output ports should be created.

1. Open the **Scoreboard.sv** file in an editor.
2. Add two new class properties:
 - **bit[3:0] sa, da;** // functional coverage properties
3. Declare a definition for a cover group (**router_cov**) immediately after the property declarations.
4. Inside the cover group
 - Create coverpoint groups based on **sa** and **da**.
 - Create cross bins based on the two sample groups (the cross coverage is the real coverage information we are looking for).

Task 3. Modify new() To Construct Coverage Object

1. In the constructor **new()**, construct **router_cov**. This is mandatory for covergroups defined inside classes. They must be constructed in the class constructor.

When done, the **covergroup** definition should look like the following:

```
class Scoreboard;
  ...
  bit[3:0] sa, da; // functional coverage properties
  covergroup router_cov;
    coverpoint sa;
    coverpoint da;
    cross sa, da;
  endgroup
  ...
endclass

function Scoreboard::new(...);
  ...
  router_cov = new();
endfunction
```

Task 4. Modify check() For Coverage

Modify the following in the **check()** function

1. Add a new **real** variable **coverage_result** (The data type must be **real** because the functional coverage results are returned as **real** values.)
You will be storing the running functional coverage % in this variable.
2. Immediately after the successful comparison of **pkt2send** and **pkt2cmp**, set the class variables **sa** and **da** to values found in the **pkt2send** object.
3. Then, update the functional coverage bin with a call to **router_cov.sample()**.
4. Make a call to **\$get_coverage()** to retrieve the updated functional coverage value and store this value in **coverage_result**.
5. Modify the **\$display()** statement that follows to also print coverage %.
6. Modify the **if** statement that follows to also trigger the **DONE** event flag if coverage reaches 100%.
7. Save and close the file.

Lab 6

Task 5. Compile and Run

1. Use make script to compile and run your program.

```
> make
```

Make sure you reach 100% coverage. Increase the number of packets if necessary. Debug any error you find.

2. Run the simulation with different seeds (other than seed value of 0 or 1).

```
> make seed=<seed_value>
```

Note that you reach 100% coverage with a different number of packets from the previous run.

3. If there are no errors, open the functional coverage html or text file and verify that each port was driven and sampled.

If the coverage report shows all combinations of input and output ports have been driven, you are done with the lab.

To view the HTML report run

```
> firefox urgReport/dashboard.html &
```

Select the groups link and follow the links for the `router_test_top.t::Scoreboard::router_cov` covergroup.

To view the report in text format run

```
> <editor> urgReport/grpinfo.txt
```

where `<editor>` is any text editor of your choice.

Part 2: Packages and Environments

Task 6. Copy Files from Labs 5/6a's Solutions directory

1. Go into the lab6b directory.

```
> cd ..\lab6b
```

2. Copy the source files in the **solutions/lab5** and **lab6a** directories into the current directory with **make** script.

```
> make copy
```

(If you choose to use your own lab files from Labs 5 and 6a, type “**make mycopy**”.)

Task 7. Create a Package for Reuse

Since many components and other classes of the testbench are reused in many tests, you can put them in a library called a **package** in SystemVerilog. Let us see how you can refine the Lab 5 test to create a reusable environment using packages.

1. Open the **test.sv** file in an editor.

Note the sections of the file. First there are some global variables – **run_for_n_packets** and **TRACE_ON**. Then there are include directives for the component files. Since these variables and components are likely to be used in different tests, for example running the same simulation with different seeds, you can put them in a reusable package.

```
program automatic test(router_io.TB rtr_io);
    int run_for_n_packets; // number of packets to test
    int TRACE_ON = 0;           // subroutine tracing control

    `include "router_test.h"
    `include "Packet.sv"
    `include "Driver.sv"
    `include "Receiver.sv"
    `include "Generator.sv"
    `include "Scoreboard.sv"

    semaphore sem[];      // prevent output port collision
    Driver     drvr[];    // driver objects
    Receiver   rcvr[];    // receiver objects
    Generator  gen;       // generator object
    Scoreboard sb;        // scoreboard object
    //continued...
```

Global Variables
and common
components –
Create Reusable
package

Lab 6

```
//test.sv continued...
initial begin
    run_for_n_packets = 2000;
    sem = new[16];
    drvr = new[16];
    rcvr = new[16];
    gen = new();
    sb = new();
    foreach (sem[i])
        sem[i] = new(1);
    for (int i=0; i<drvr.size(); i++)
        drvr[i] = new($psprintf("drvr[%0d]", i), ., sem,
                      gen.out_box[i], sb.driver_mbox, rtr_io);
    for (int i=0; i<rcvr.size(); i++)
        rcvr[i] = new($psprintf("rcvr[%0d]", i), .,
                      sb.receiver_mbox, rtr_io);

    reset();
    gen.start();
    sb.start();
    foreach(drvr[i])
        drvr[i].start();
    foreach(rcvr[i])
        rcvr[i].start();

    wait(sb.DONE.triggered);
end

task reset();
    if (TRACE_ON) $display("[TRACE] %t %m", $realtime);
    rtr_io.reset_n = 1'b0;
    rtr_io.cb.frame_n <= '1;
    rtr_io.cb.valid_n <= '1;
    repeat(2) @rtr_io.cb;
    rtr_io.cb.reset_n <= 1'b1;
    repeat(15) @(rtr_io.cb);
endtask
endprogram
```

```
graph TD
    A[Build the Environment  
(construct Testbench components)] --> B[Reset DUT]
    B --> C[Start the TB components running concurrently]
    C --> D[Wait for end-of-test]
    D --> E[Reset DUT]
    E --> F[Start the TB components running concurrently]
    F --> G[Wait for end-of-test]
```

The flowchart illustrates the testbench execution steps. It starts with 'Build the Environment (construct Testbench components)', followed by 'Reset DUT'. This leads to 'Start the TB components running concurrently', which then leads to 'Wait for end-of-test'. Arrows indicate a feedback loop from 'Wait for end-of-test' back to 'Reset DUT', 'Start the TB components running concurrently', and finally 'Wait for end-of-test' again.

2. Close the file.
3. Edit the file **router_test_pkg.sv** in an editor.
4. Declare a new package **router_test_pkg**.
 - **package router_test_pkg;**

Lab 6-8

Environments, Packages, Functional Coverage
Synopsys SVTB Workshop

5. Add an int property **run_for_n_packets** and initialize it to 0.
 - `int run_for_n_packets = 0;`
 6. Add the **TRACE_ON** variable and initialize it to 0. Packages are top-level name spaces in SystemVerilog and hence global variables should go in packages.
 - `int TRACE_ON = 0;`
 7. Include the following files to create the package.
 - ``include "router_test.h"`
 - ``include "Packet.sv"`
 - ``include "Driver.sv"`
 - ``include "Receiver.sv"`
 - ``include "Generator.sv"`
 - ``include "Scoreboard.sv"`
 - ``include "Environment.sv"`
- We will use the Environment class after Task 9.
8. Complete the package definition.
 9. Save and close the file.

Task 8. Use a Package in the Test Program

1. Open the file **test.sv** in an editor.
2. Delete the variable definitions and the include directives. Do not delete the component declarations and the initial block.
3. Import the package created in Task 2. Refer to your slides for the syntax. By importing the package everything in the package is visible in the program scope.
4. Save and close the file.

Task 9. Compile and Run

1. Use **make** script to compile and run your program.

```
> make
```

This simulation run is similar to the one you ran in Task 5. You should see the simulation stop after reaching 100% coverage or after 2000 packets have been transmitted. You may need to increase the number of packets transmitted to reach 100% coverage.

Task 10. Create a Test Environment for Reuse

The test environment is made up of components and their interconnections. The idea of an environment is to provide a test infrastructure with “knobs” that can be changed to create individual tests. These knobs can consist of extended classes to change behavior of existing classes, configuration variables that can be modified, randomized, etc. This also takes advantage of SystemVerilog’s built-in randomization capability. The idea is to be able to set and control these knobs without having to directly modify the code in the environment and its components. By creating random configurations you can test many more configurations and test conditions than you can with directed tests. The components like drivers, monitors etc. can also be constructed, started and stopped in a standardized manner.

1. Open the **Environment.sv** file in an editor. The complete environment has been coded for you.
2. Note the random variable **run_for_n_packets**. By putting random variables in the environment, or in a configuration object inside the environment, you can randomize the test configuration. Note the **constraint** block definition.
3. Note the methods called **build()**, **reset()**, **configure()**, **start()**, **wait_for_end()**. These correspond to the usual steps of most simulations for functional verification.
4. The **configure()** method is used to configure the test environment. Here it simply randomizes the **run_for_n_packets** variable. In a more complex verification environment, you would have many “knobs” in your test to configure. For example, knobs that control how many drivers or monitors to run the test with, or, what types of packets to drive. All these could be randomized during the test configuration step.
5. The **run()** method calls **build()**, then **reset()**, followed by **start()** and **wait_for_end()**. In the **build()** you construct the components needed by the test. The **reset()** is used to reset the DUT. The **start()** method will start all the components by calling the **start()** methods of each component constructed in the **build()** method. The **wait_for_end()** method waits for end-of-test conditions. You may think of these methods that **run()** calls as implementing the phases of the test run.

```

class Environment;
  string name;
  rand int run_for_n_packets; // number of packets to test
  virtual router_io.TB rtr_io;

  semaphore sem[];      // prevent output port collision
  Driver drvr[];       // driver objects
  Receiver rcvr[];     // receiver objects
  Generator gen[];     // generator objects
  Scoreboard sb;        // scoreboard object

  constraint valid {
    this.run_for_n_packets inside { [1500:2500] };
  }

  extern function new(string name = "Env",
                      virtual router_io.TB rtr_io);
  extern virtual task run();
  extern virtual function void configure();
  extern virtual function void build();
  extern virtual task start();
  extern virtual task wait_for_end();
  extern virtual task reset();

endclass: Environment

function Environment::new(string name = "Env", virtual
router_io.TB rtr_io);
  if (TRACE_ON) $display("[TRACE] %t %s:%m", $realtime, name);
  this.name = name;
  this.rtr_io = rtr_io;
endfunction: new

task Environment::run();
  if (TRACE_ON) $display("[TRACE] %t %s:%m", $realtime,
this.name);
  this.build();
  this.reset();
  this.start();
  this.wait_for_end();
endtask: run

function void Environment::configure();
  if (TRACE_ON) $display("[TRACE] %t %s:%m", $realtime, this.name);
  this.randomize();
endfunction: configure

```

"Run" the
Environment

Lab 6

```
//Class Environment continued...
function void Environment::build();
    if (TRACE_ON) $display("[TRACE] %t %s:%m", $realtime, this.name);
    if(this.run_for_n_packets == 0) this.run_for_n_packets = 2000;
    this.sem = new[16];
    this.drvr = new[16];
    this.rcvr = new[16];
    this.gen = new[16];
    this.sb = new();
    foreach (this.sem[i])
        this.sem[i] = new(1);
    foreach (this.gen[i])
        this.gen[i] = new($sformatf("gen[%0d]", i));
    foreach (this.drvr[i])
        this.drvr[i] = new($psprintf("drvrv[%0d]", i), i, this.sem,
                           this.gen.out_box[i], this.sb.driver_mbox, this.rtr_io);
    foreach (this.rcvr[i])
        this.rcvr[i] = new($psprintf("rcvrv[%0d]", i), i,
                           this.sb.receiver_mbox, this.rtr_io);
endfunction: build

task Environment::reset();
    if (TRACE_ON) $display("[TRACE] %t %s:%m", $realtime, this.name);
    this.rtr_io.reset_n <= 1'b0;
    this.rtr_io.cb.frame_n <= '1;
    this.rtr_io.cb.valid_n <= '1;
    repeat(2) @rtr_io.cb;
    this.rtr_io.cb.reset_n <= 1'b1;
    repeat(15) @(this.rtr_io.cb);
endtask: reset

task Environment::start();
    if (TRACE_ON) $display("[TRACE] %t %s:%m", $realtime, this.name);
    this.sb.start();
    foreach(this.gen[i])
        this.gen[i].start();
    foreach(this.drvr[i])
        this.drvr[i].start();
    foreach(this.rcvr[i])
        this.rcvr[i].start();
endtask: start

task Environment::wait_for_end();
    if (TRACE_ON) $display("[TRACE] %t %s:%m", $realtime, this.name);
    wait(this.sb.DONE.triggered);
endtask: wait_for_end
```

```
graph TD; A[Build the Environment<br>(construct components)] --> B[Reset DUT]; B --> C[Start the components]; C --> D[Wait for end-of-test];
```

Lab 6-12

Environments, Packages, Functional Coverage
Synopsys SVTB Workshop

Task 11. Use Environment in the Test Program

1. Open the file `test.sv` in an editor.

Note that in the `initial` block you perform the following steps to run your test.

- The variables are set (configured).
- The components constructed (built).
- The DUT is reset.
- The components are started. This starts the simulation traffic.
- Finally the test waits for the end-of-test condition – the `DONE` event.

These are common test tasks or **phases** that you will now encapsulate into an Environment class.

2. Delete the handle declarations for the components.
3. Immediately after the import statement declare a handle to the Environment class. Call it `env`.
4. Delete all code in the `initial` block.
5. In the `initial` block construct the Environment object. Remember to pass the correct arguments to the constructor.
6. Call the `configure()` method of the Environment. This is one way to control the knobs of the test. In our test this randomizes the Environment's `run_for_n_packets` variable.
7. Update the `run_for_n_packets` (defined in the package, available in the program) to environment's `run_for_n_packets`.
8. "Run" the test by calling the `run()` method of the Environment. The Environment will now handle all the phases of running the simulation that you performed in the `program` in Lab 5.
9. Delete the `reset()` task.

Lab 6

- When done, the test program should look like the following:

```
program automatic test(router_io.TB rtr_io);
import router_test_pkg::*;
Environment env;
initial begin
    env = new("env", rtr_io);
    env.configure();
    run_for_n_packets = env.run_for_n_packets;
    env.run();
end
endprogram: test
```

Task 12. Compile and Run

- Use make script to compile and run your program.

> **make**

Question 1. How is this run different from the run in Task 8? Why?

.....
.....
.....

- Rerun simulation with a different seed.

> **make run seed=<seed_value>**

Question 2. How did the seed affect the results? Why?

.....
.....

Congratulations, you completed Lab 6!

Answers / Solutions

Task 12. Compile and Run

Question 1. How is this run different from the run in Task 8? Why?

- The number of packets transmitted in the two tests was not the same. The Environment configuration that was performed when you called `env.configure()`, randomized the `run_for_n_packets` variable. This was used by the test. The constraint for this random variable is defined in the `Environment` class.
- The number of packets required to reach 100% coverage also changed. Since the structure of the two test environments is different, the randomization of packets was not the same between the two tests. So the number of packets needed to reach 100% coverage was also different. This is a good example of how changing the test environment affects the constraint solver. If you change the environment after fixing a bug in the DUT you may not be able to reproduce the bug.

Question 2. How did the seed affect the results? Why?

- The number of packets required to reach 100% coverage changed. This is a good example of how changing the simulation random seed affects the constraint solver and thus affects the time to coverage closure.

Scoreboard.sv Solution Lab6a:

```
`ifndef INC_SCOREBOARD_SV
`define INC_SCOREBOARD_SV
class Scoreboard;
    string name;      // unique identifier
    event DONE;       // flag to indicate goal reached
    Packet refPkt[$]; // reference Packet array
    Packet pkt2send; // Packet object from Drivers
    Packet pkt2cmp; // Packet object from Receivers
    pkt_mbox driver_mbox; // mailbox for Packet objects from Drivers
    pkt_mbox receiver_mbox; // mailbox for Packet objects from Receivers
    bit[3:0] sa, da; // functional coverage properties
    int run_for_n_packets; //how many packets

    covergroup router_cov;
        coverpoint sa ;
        coverpoint da ;
        cross sa, da;
    endgroup

    extern function new(string name = "Scoreboard",
                        pkt_mbox driver_mbox = null, receiver_mbox = null);
    extern virtual task start();
    extern virtual function void check();
endclass

function Scoreboard::new(string name, pkt_mbox driver_mbox,
receiver_mbox);
    if (TRACE_ON) $display("[TRACE] %0t %s:%m", $time, name);
    this.name = name;
    if (driver_mbox == null) driver_mbox = new();
    this.driver_mbox = driver_mbox;
    if (receiver_mbox == null) receiver_mbox = new();
    this.receiver_mbox = receiver_mbox;
    router_cov = new();
endfunction
```

Continued...

```
task Scoreboard::start();
    if (TRACE_ON) $display("[TRACE] %0t %s:%m", $time, name);
    fork
        forever begin
            receiver_mbox.get(pkt2cmp);
            while (driver_mbox.num()) begin
                Packet pkt;
                driver_mbox.get(pkt);
                refPkt.push_back(pkt);
            end
            check();
        end
        join_none
    endtask

function void Scoreboard::check();
    int index[$];
    string message;
    static int pkts_checked = 0;
    real coverage_result;

    if (TRACE_ON) $display("[TRACE] %0t %s:%m", $time, name);
    index = refPkt.find_first_index() with (item.da == pkt2cmp.da);
    if (index.size() <= 0) begin
        $display("\n%m\n[ERROR] %0t %s not found in Reference Queue\n",
$time, pkt2cmp.name);
        pkt2cmp.display("ERROR");
        $finish;
    end
    pkt2send = refPkt[index[0]];
    refPkt.delete(index[0]);
    if (!pkt2send.compare(pkt2cmp, message)) begin
        $display("\n%m\n[ERROR] %0t Packet #%0d %s\n", $time, pkts_checked,
message);
        pkt2send.display("ERROR");
        pkt2cmp.display("ERROR");
        $finish;
    end
    this.sa = pkt2send.sa;
    this.da = pkt2send.da;
    router_cov.sample();
    coverage_result = $get_coverage();
    $display("[NOTE] %0t Packet #%0d %s coverage = %3.2f", $time,
pkts_checked++, message, coverage_result);
    if ((pkts_checked >= run_for_n_packets) || (coverage_result == 100))
        ->DONE;
endfunction: check
`endif
```

router test pkg.sv Solution Lab 6b:

```
package router_test_pkg;

int run_for_n_packets = 0;
int TRACE_ON = 0;

`include "router_test.h"
`include "Packet.sv"
`include "Driver.sv"
`include "Receiver.sv"
`include "Generator.sv"
`include "Scoreboard.sv"
`include "Environment.sv"

endpackage: router_test_pkg
```

test.sv Solution Lab 6b:

```
program automatic test(router_io.TB rtr_io);
import router_test_pkg::*;

Environment env;

initial begin
    env = new("env", rtr_io);
    env.configure();
    run_for_n_packets = env.run_for_n_packets;
    env.run();
end

endprogram: test
```